

Mathematische Methoden in der Physik

Bonus Unterlagen : Aufgaben und Lösungen

Hinweis : Diese Texte wurden mit Hilfe von WOLFRAM MATHEMATICA 12 (Wolfram Research, Inc.) erstellt.

Viele in diesen Aufgaben gestellten Probleme können mit *Mathematica* direkt durch spezielle Befehle (wie etwa Limit oder Series) gelöst werden. Da es uns hier aber auf das entsprechende Hintergrundwissen ankommt, werden wir meist ausführlicher vorgehen und erst in späteren Rechnungen oder zur Kontrolle diese Befehle verwenden.

Kapitel 8. Basissysteme krummliniger Koordinaten

8.1

Man berechne $(ds)^2$ für folgende Koordinatensysteme:

$$(a) \quad x_1 = \frac{1}{2} (u^2 - v^2), \quad x_2 = u v,$$

$x_3 = x_3$ (parabolische Zylinderkoordinaten);

$$(b) \quad x_1 = a \cosh u \cos v, \quad x_2 = a \sinh u \sin v,$$

$x_3 = x_3$ (elliptische Zylinderkoordinaten);

$$(c) \quad x_1 = r \sin \theta \cos \varphi, \quad x_2 = r \sin \theta \sin \varphi,$$

$x_3 = r \cos \theta$ (sphärische Polarkoordinaten).

Lösungsweg

(a)

Es ist zum Beispiel

$$d \mathbf{r}_u = \left| \frac{\partial \mathbf{r}}{\partial u} \right| du \mathbf{e}_u$$

und also

$$\text{In[]:= } \mathbf{r} = \{ (u^2 - v^2) / 2, u v, x_3 \};$$

Wir berechnen

In[*]:= **D[r, u]**

Out[*]= {u, v, 0}

In[*]:= **D[r, v]**

Out[*]= {-v, u, 0}

In[*]:= **D[r, x3]**

Out[*]= {0, 0, 1}

In[*]:= **dru = Sqrt[D[r, u].D[r, u]] du**

Out[*]= $du \sqrt{u^2 + v^2}$

In[*]:= **drv = Sqrt[D[r, v].D[r, v]] dv**

Out[*]= $dv \sqrt{u^2 + v^2}$

In[*]:=

drx3 = Sqrt[D[r, x3].D[r, x3]] dx3

Out[*]= dx3

Ist das Koordinatensystem ein orthogonales? Wir kontrollieren:

In[*]:= **D[r, u].D[r, v]**

Out[*]= 0

In[*]:= **D[r, u].D[r, x3]**

Out[*]= 0

In[*]:= **D[r, v].D[r, x3]**

Out[*]= 0

In Ordnung. Es ist also $(ds)^2$:

In[*]:= **dru^2 + drv^2 + drx3^2**

Out[*]= $dx3^2 + du^2 (u^2 + v^2) + dv^2 (u^2 + v^2)$

In[*]:= **ClearAll["Global`*"];**

(b)

Es ist zum Beispiel

$$d\mathbf{r}_u = \left| \frac{\partial \mathbf{r}}{\partial u} \right| du \mathbf{e}_u$$

und also

In[*]:= **r = {a Cosh[u] Cos[v], a Sinh[u] Sin[v], x3};**

Wir berechnen

In[*]:= **D[r, u]**

Out[*]= {a Cos[v] Sinh[u], a Cosh[u] Sin[v], 0}

```
In[*]:= D[r, v]
```

```
Out[*]:= {-a Cosh[u] Sin[v], a Cos[v] Sinh[u], 0}
```

```
In[*]:= D[r, x3]
```

```
Out[*]:= {0, 0, 1}
```

```
In[*]:= dru = Simplify[Sqrt[D[r, u].D[r, u]]] du
```

```
Out[*]:= du sqrt(a^2 (Cosh[u]^2 Sin[v]^2 + Cos[v]^2 Sinh[u]^2))
```

```
In[*]:= drv = Simplify[Sqrt[D[r, v].D[r, v]]] dv
```

```
Out[*]:= dv sqrt(a^2 (Cosh[u]^2 Sin[v]^2 + Cos[v]^2 Sinh[u]^2))
```

```
In[*]:= drx3 = Sqrt[D[r, x3].D[r, x3]] dx3
```

```
Out[*]:= dx3
```

Ist das Koordinatensystem ein orthogonales? Wir kontrollieren:

```
In[*]:= D[r, u].D[r, v]
```

```
Out[*]:= 0
```

```
In[*]:= D[r, u].D[r, x3]
```

```
Out[*]:= 0
```

```
In[*]:= D[r, v].D[r, x3]
```

```
Out[*]:= 0
```

In Ordnung. Es ist also $(ds)^2$:

Wir berechnen

```
In[*]:= ds2 = Simplify[dru^2 + drv^2 + drx3^2]
```

```
Out[*]:= dx3^2 - 1/2 a^2 (du^2 + dv^2) Cos[2 v] + 1/2 a^2 (du^2 + dv^2) Cosh[2 u]
```

Oder, anders ausgedrückt

```
In[*]:= Simplify[ds2 /. {Cos[2 v] -> 1 - 2 Sin[v]^2,
```

```
Cosh[2 u] -> 1 + 2 Sinh[u]^2}]
```

```
Out[*]:= dx3^2 + a^2 (du^2 + dv^2) Sin[v]^2 + a^2 (du^2 + dv^2) Sinh[u]^2
```

```
In[*]:= ClearAll["Global`*"];
```

(c)

Es ist zum Beispiel

$$d\mathbf{r}_u = \left| \frac{\partial \mathbf{r}}{\partial u} \right| du \mathbf{e}_u$$

und also

```
In[*]:= Clear[r]
```

```
In[*]:= rvec = {r Sin[θ] Cos[φ], r Sin[θ] Sin[φ], r Cos[θ]};
```

Wir berechnen

```
In[*]:= D[rvec,  $\theta$ ]
```

```
Out[*]:= {r Cos[ $\theta$ ] Cos[ $\varphi$ ], r Cos[ $\theta$ ] Sin[ $\varphi$ ], -r Sin[ $\theta$ ] }
```

```
In[*]:= D[rvec,  $\varphi$ ]
```

```
Out[*]:= {-r Sin[ $\theta$ ] Sin[ $\varphi$ ], r Cos[ $\varphi$ ] Sin[ $\theta$ ], 0 }
```

```
In[*]:= D[rvec, r]
```

```
Out[*]:= {Cos[ $\varphi$ ] Sin[ $\theta$ ], Sin[ $\theta$ ] Sin[ $\varphi$ ], Cos[ $\theta$ ] }
```

```
In[*]:= dr $\theta$  = Simplify[Sqrt[D[rvec,  $\theta$ ].D[rvec,  $\theta$ ]]] d $\theta$ 
```

```
Out[*]:= d $\theta$   $\sqrt{r^2}$ 
```

oder auch:

```
In[*]:= dr $\theta$  = r d $\theta$ 
```

```
Out[*]:= d $\theta$  r
```

```
In[*]:= dr $\varphi$  = Simplify[Sqrt[D[rvec,  $\varphi$ ].D[rvec,  $\varphi$ ]]] d $\varphi$ 
```

```
Out[*]:= d $\varphi$   $\sqrt{r^2 \text{Sin}[\theta]^2}$ 
```

oder auch:

```
In[*]:= dr $\varphi$  = r Sin[ $\theta$ ] d $\varphi$ 
```

```
Out[*]:= d $\varphi$  r Sin[ $\theta$ ]
```

```
In[*]:= drr = Simplify[Sqrt[D[rvec, r].D[rvec, r]]] dr
```

```
Out[*]:= dr
```

Ist das Koordinatensystem ein orthogonales? Wir kontrollieren:

```
In[*]:= D[rvec,  $\theta$ ].D[rvec,  $\varphi$ ]
```

```
Out[*]:= 0
```

```
In[*]:= Simplify[D[rvec,  $\theta$ ].D[rvec, r]]
```

```
Out[*]:= 0
```

```
In[*]:= D[rvec,  $\varphi$ ].D[rvec, r]
```

```
Out[*]:= 0
```

In Ordnung. Es ist also $(ds)^2$:

```
In[*]:= ds2 = Simplify[dr $\theta$ ^2 + dr $\varphi$ ^2 + drr^2]
```

```
Out[*]:= dr^2 + d $\theta$ ^2 r^2 + d $\varphi$ ^2 r^2 Sin[ $\theta$ ]^2
```

```
In[*]:= ClearAll["Global`*"];
```

8.2

Berechnen Sie **div v** für folgende Fälle in der kartesischen Basis und in der Basis der Zylinderkoordinaten (Angegebene Einheitsvektoren aus der Zylinderkoordinatenbasis):

(a) $\mathbf{v} = \mathbf{e}_\rho$

$$(b) \mathbf{v} = \mathbf{e}_\varphi$$

$$(c) \mathbf{v} = \rho \mathbf{e}_\rho + x_3 \mathbf{e}_3$$

$$(d) \mathbf{v} = \rho \mathbf{e}_\varphi$$

Lösungsweg

(a)

Der Einheitsvektor $\mathbf{e}_\rho$ im Zylinderkoordinatensystem hat in der kartesischen Basis die Form (siehe Buch (8.10))

```
In[ ]:= vec = {Cos[phi], Sin[phi], 0}
```

```
Out[ ]:= {Cos[phi], Sin[phi], 0}
```

wobei

```
In[ ]:= phi = ArcTan[y / x]
```

```
Out[ ]:= ArcTan[ $\frac{y}{x}$ ]
```

gilt. Es ist also

```
In[ ]:= divv = D[vec[[1]], x] + D[vec[[2]], y] + D[vec[[3]], z]
```

```
Out[ ]:=  $\frac{1}{x \sqrt{1 + \frac{y^2}{x^2}}}$ 
```

oder, umgeformt: $1/r$.

Das geht in Zylinderkoordinaten schneller. Dort ist

```
In[ ]:= vec = {1, 0, 0}
```

```
Out[ ]:= {1, 0, 0}
```

und $\mathbf{div} \mathbf{v} = \nabla \cdot \mathbf{v}$ hat (entsprechend (8.35) im Buch) die Form

```
In[ ]:=  $\frac{1}{r} \left[ \frac{\partial}{\partial r} (r v_1) + \frac{\partial}{\partial \varphi} v_2 + \frac{\partial}{\partial z} (r v_{31}) \right]$ 
```



Es ergibt sich daraus sofort $\mathbf{div} \mathbf{v} = 1/r$.

```
In[ ]:= ClearAll["Global`*"];
```

(b)

In (a) haben wir die notwendigen Teilschritte ausführlich besprochen. Nun verwenden wir einfach die in Mathematica eingebauten Befehle.

```
In[ ]:= Needs["VectorAnalysis`"];
```

```
In[ ]:= SetCoordinates[Cartesian[x, y, z]];
```

Der Einheitsvektor $\mathbf{e}_\varphi$ im Zylinderkoordinatensystem hat in der kartesischen Basis die Form

```
In[ ]:= vec = {-Sin[phi], Cos[phi], 0}
```

```
Out[ ]:= {-Sin[phi], Cos[phi], 0}
```

wobei

```
In[ ]:= phi = ArcTan[y / x]
```

```
Out[ ]:= ArcTan[ $\frac{y}{x}$ ]
```

gilt. Es ist also

```
In[ ]:= divv = D[vec[[1]], x] + D[vec[[2]], y] + D[vec[[3]], z]
```

```
Out[ ]:= 
$$-\frac{y}{x^2 \left(1 + \frac{y^2}{x^2}\right)^{3/2}} - \frac{y^3}{x^4 \left(1 + \frac{y^2}{x^2}\right)^{3/2}} + \frac{y}{x^2 \sqrt{1 + \frac{y^2}{x^2}}}$$

```

Wir hätten auch einfach schreiben können

```
In[ ]:= Div[vec]
```

```
Out[ ]:= 
$$-\frac{y}{x^2 \left(1 + \frac{y^2}{x^2}\right)^{3/2}} - \frac{y^3}{x^4 \left(1 + \frac{y^2}{x^2}\right)^{3/2}} + \frac{y}{x^2 \sqrt{1 + \frac{y^2}{x^2}}}$$

```

Vereinfacht gibt das

```
In[ ]:= Simplify[divv]
```

```
Out[ ]:= 0
```

Das geht in Zylinderkoordinaten schneller:

```
In[ ]:= Clear[r, phi, z]; SetCoordinates[Cylindrical[r, phi, z]];
```

Damit haben wir uns auf das Zylinderkoordinatensystem festgelegt. Es ergibt sich sofort

```
In[ ]:= Div[{0, 1, 0}]
```

```
Out[ ]:= 0
```

```
In[ ]:= ClearAll["Global`*"];
```

(c)

In (a) haben wir die notwendigen Teilschritte ausführlich besprochen. Hier verwenden wir einfach die in Mathematica eingebauten Befehle.

```
In[ ]:= Needs["VectorAnalysis`"];
```

```
In[ ]:= SetCoordinates[Cartesian[x, y, z]];
```

Der Einheitsvektor $\mathbf{e}_\rho$ im Zylinderkoordinatensystem hat in der kartesischen Basis die Form $\{\cos[\phi], \sin[\phi], 0\}$ und daher ist

```
In[ ]:= Clear[r, phi, z]; vec = {r Cos[phi], r Sin[phi], z}
```

```
Out[ ]:= {r Cos[phi], r Sin[phi], z}
```

wobei

```
In[ ]:= phi = ArcTan[y / x]; r = Sqrt[x^2 + y^2];
```

gilt. Es ist also

In[]:= **Simplify**[Div[vec]]

$$\text{Out[]}= 1 + \frac{2x\sqrt{1 + \frac{y^2}{x^2}}}{\sqrt{x^2 + y^2}}$$

Vereinfacht gibt das 3.

Das geht in Zylinderkoordinaten schneller:

In[]:= **Clear**[r, phi, z]; **SetCoordinates**[Cylindrical[r, phi, z]];

Damit haben wir uns auf das Zylinderkoordinatensystem festgelegt. Es ergibt sich sofort

In[]:= **Div**[{r, 0, z}]

Out[]:= 3

In[]:= **ClearAll**["Global`*"];

(d)

In (a) haben wir die notwendigen Teilschritte ausführlich besprochen. Hier verwenden wir einfach die in Mathematica eingebauten Befehle.

In[]:= **Needs**["VectorAnalysis`"];

In[]:= **SetCoordinates**[Cartesian[x, y, z]];

Der Einheitsvektor $\mathbf{e}_\phi$ im Zylinderkoordinatensystem hat in der kartesischen Basis die Form $\{-\sin[\phi], \cos[\phi], 0\}$ und daher ist

In[]:= **Clear**[r, phi, z]; **vec** = $\{-r \sin[\phi], r \cos[\phi], 0\}$

Out[]:= $\{-r \sin[\phi], r \cos[\phi], 0\}$

wobei

In[]:= **phi** = **ArcTan**[y / x]

Out[]:= $\text{ArcTan}\left[\frac{y}{x}\right]$

gilt. Es ist also

In[]:= **Simplify**[Div[vec]]

Out[]:= 0

Das geht in Zylinderkoordinaten schneller:

In[]:= **Clear**[r, phi, z]; **SetCoordinates**[Cylindrical[r, phi, z]];

Damit haben wir uns auf das Zylinderkoordinatensystem festgelegt. Es ergibt sich sofort

In[]:= **Div**[{0, r, 0}]

Out[]:= 0

```
In[ ]:= ClearAll["Global`*"];
```

```
In[ ]:= ClearAll["Global`*"];
```

8.3

Man zeige, dass für $\rho \neq 0$ die Laplace-Gleichung

$\Delta \Phi = 0$ erfüllt ist für

(a) $\Phi(\rho, \varphi, x_3) = \rho^n \sin n \varphi$ ($n = \text{ganzzahlig}$),

(b) $\Phi(\rho, \varphi, x_3) = \rho^n \cos n \varphi$ ($n = \text{ganzzahlig}$).

Lösungsweg

Die Laplace-Gleichung in Zylinderkoordinaten ist in (M.8.1.2) angegeben.

```
In[ ]:= Needs["VectorAnalysis`"];
```

```
In[ ]:= SetCoordinates[Cylindrical[\rho, \varphi, z]];
```

```
In[ ]:= Laplacian[f[\rho, \varphi, z]]
```

$$\text{Out[]:= } \frac{\rho f^{(0,0,2)}[\rho, \varphi, z] + \frac{f^{(0,2,0)}[\rho, \varphi, z]}{\rho} + f^{(1,0,0)}[\rho, \varphi, z] + \rho f^{(2,0,0)}[\rho, \varphi, z]}{\rho}$$

```
In[ ]:= Laplacian[\rho^n Sin[n \varphi]]
```

$$\text{Out[]:= } 0$$

```
In[ ]:= Laplacian[\rho^n Cos[n \varphi]]
```

$$\text{Out[]:= } 0$$

Die Laplace-Gleichung ist also erfüllt.

Das verwundert übrigens nicht, da die Funktionen Real- und Imaginärteil der komplexen, analytischen Funktion z^n sind und eben daher die Gleichung erfüllen (vgl. Kap. 19).

```
In[ ]:= ClearAll["Global`*"];
```

```
In[ ]:= ClearAll["Global`*"];
```

8.4

Bestimmen Sie $\nabla \Phi$ und $\Delta \Phi$ für

(a) $\Phi(\rho, \varphi, x_3) = \sin \varphi$

(b) $\Phi(r, \theta, \varphi) = r^2 \sin \theta$

Lösungsweg

(a)

Gradient und Laplace-Gleichung in Zylinderkoordinaten sind in (M.8.12) angegeben.

```
In[ ]:= Needs["VectorAnalysis`"];
```

In[]:= SetCoordinates[Cylindrical[ρ, φ, z]];

In[]:= Grad[f[ρ, φ, z]]

$$\text{Out[]} = \left\{ f^{(1,0,0)}[\rho, \varphi, z], \frac{f^{(0,1,0)}[\rho, \varphi, z]}{\rho}, f^{(0,0,1)}[\rho, \varphi, z] \right\}$$

In[]:= Laplacian[f[ρ, φ, z]]

$$\text{Out[]} = \frac{\rho f^{(0,0,2)}[\rho, \varphi, z] + \frac{f^{(0,2,0)}[\rho, \varphi, z]}{\rho} + f^{(1,0,0)}[\rho, \varphi, z] + \rho f^{(2,0,0)}[\rho, \varphi, z]}{\rho}$$

Wir berechnen damit also

In[]:= Grad[Sin[φ]]

$$\text{Out[]} = \left\{ 0, \frac{\cos[\varphi]}{\rho}, 0 \right\}$$

In[]:= Laplacian[Sin[φ]]

$$\text{Out[]} = -\frac{\sin[\varphi]}{\rho^2}$$

In[]:= ClearAll["Global`*"];

(b)

Gradient und Laplace-Gleichung in Zylinderkoordinaten sind in (M.8.12) angegeben.

In[]:= Needs["VectorAnalysis`"];

In[]:= SetCoordinates[Spherical[r, θ, φ]];

In[]:= Grad[f[r, θ, φ]]

$$\text{Out[]} = \left\{ f^{(1,0,0)}[r, \theta, \varphi], \frac{f^{(0,1,0)}[r, \theta, \varphi]}{r}, \frac{\csc[\theta] f^{(0,0,1)}[r, \theta, \varphi]}{r} \right\}$$

In[]:= Laplacian[f[r, θ, φ]]

$$\text{Out[]} = \frac{1}{r^2} \csc[\theta] \left(\csc[\theta] f^{(0,0,2)}[r, \theta, \varphi] + \cos[\theta] f^{(0,1,0)}[r, \theta, \varphi] + \sin[\theta] f^{(0,2,0)}[r, \theta, \varphi] + 2 r \sin[\theta] f^{(1,0,0)}[r, \theta, \varphi] + r^2 \sin[\theta] f^{(2,0,0)}[r, \theta, \varphi] \right)$$

Wir berechnen damit also

In[]:= Grad[r^2 Sin[θ]]

$$\text{Out[]} = \{ 2 r \sin[\theta], r \cos[\theta], 0 \}$$

In[]:= Laplacian[r^2 Sin[θ]]

$$\text{Out[]} = \frac{\csc[\theta] \left(r^2 \cos[\theta]^2 + 5 r^2 \sin[\theta]^2 \right)}{r^2}$$

In[]:= Simplify[Laplacian[r^2 Sin[θ]]]

$$\text{Out[]} = (5 + \cot[\theta]^2) \sin[\theta]$$

```
In[ ]:= ClearAll["Global`*"];
```

```
In[ ]:= ClearAll["Global`*"];
```

8.5

Man zeige für sphärische Polarkoordinaten die Relationen

$$\mathbf{e}_r = \dot{\theta} \mathbf{e}_\theta + \sin \theta \dot{\varphi} \mathbf{e}_\varphi$$

$$\mathbf{e}_\theta = -\dot{\theta} \mathbf{e}_r + \cos \theta \dot{\varphi} \mathbf{e}_\varphi,$$

$$\mathbf{e}_\varphi = -\sin \theta \dot{\varphi} \mathbf{e}_r - \cos \theta \dot{\varphi} \mathbf{e}_\theta.$$

Lösungsweg

Diese Beziehungen werden im Beispiel auf Seite 302 verwendet.

Laut (8.16) sind die Einheitsvektoren in Kugelkoordinaten (=sphärische Polarkoordinaten), ausgedrückt im Basissystem der kartesischen Koordinaten:

```
In[ ]:= erho = {Sin[th[t]] Cos[phi[t]], Sin[th[t]] Sin[phi[t]], Cos[th[t]]};
```

```
eth = {Cos[th[t]] Cos[phi[t]], Cos[th[t]] Sin[phi[t]], -Sin[th[t]]};
```

```
ephi = {-Sin[phi[t]], Cos[phi[t]], 0};
```

Wir haben dabei die Abhängigkeit nach der Zeit explizit angegeben, um die nachfolgende Differenziation vorzubereiten.

```
In[ ]:= derhot = D[erho, t]
```

```
Out[ ]:= {-Sin[phi[t]] Sin[th[t]] phi'[t] + Cos[phi[t]] Cos[th[t]] th'[t],
          Cos[phi[t]] Sin[th[t]] phi'[t] + Cos[th[t]] Sin[phi[t]] th'[t], -Sin[th[t]] th'[t]}
```

```
In[ ]:= detht = D[eth, t]
```

```
Out[ ]:= {-Cos[th[t]] Sin[phi[t]] phi'[t] - Cos[phi[t]] Sin[th[t]] th'[t],
          Cos[phi[t]] Cos[th[t]] phi'[t] - Sin[phi[t]] Sin[th[t]] th'[t], -Cos[th[t]] th'[t]}
```

```
In[ ]:= dephit = D[ephi, t]
```

```
Out[ ]:= {-Cos[phi[t]] phi'[t], -Sin[phi[t]] phi'[t], 0}
```

Die gesuchten Komponenten erhalten wir durch Projektion:

```
In[ ]:= Simplify[derhot.eth]
```

```
Out[ ]:= th'[t]
```

```
In[ ]:= Simplify[derhot.ephi]
```

```
Out[ ]:= Sin[th[t]] phi'[t]
```

```
In[ ]:= Simplify[detht.erho]
```

```
Out[ ]:= -th'[t]
```

```
In[ ]:= Simplify[detht.ephi]
```

```
Out[ ]:= Cos[th[t]] phi'[t]
```

```
In[ ]:= Simplify[dephit.erro]
```

```
Out[ ]:= -Sin[th[t]] phi'[t]
```

```
In[ ]:= Simplify[dephit.eth]
```

```
Out[ ]:= -Cos[th[t]] phi'[t]
```

Wegen der Konstanz der Einheitsvektoren verschwindet jeweils die Projektion des Einheitsvektors auf seine Ableitung.

```
In[ ]:= ClearAll["Global`*"];
```

```
In[ ]:= ClearAll["Global`*"];
```

8.6

Zeigen Sie für die Koordinaten (elliptische Zylinderkoordinaten)

$$x_1 = a \cosh u_1 \cos u_2, \quad u_1 \geq 0$$

$$x_2 = a \sinh u_1 \sin u_2, \quad 0 \leq u_2 < 2\pi$$

$$x_3 = x_3, \quad -\infty < x_3 < +\infty$$

(a) ihre Orthogonalität und

(b) berechnen Sie ihr Volumenelement dV .

Lösungsweg

Der Ortsvektor hat die Form

```
In[ ]:= rvec = {a * Cosh[u1] * Cos[u2],  
               a * Sinh[u1] * Sin[u2], z}
```

```
Out[ ]:= {a Cos[u2] Cosh[u1], a Sin[u2] Sinh[u1], z}
```

Zur Berechnung der Norm bereiten wir eine Funktion vor:

```
In[ ]:= Norm[vector_] := Simplify[Sqrt[vector.vector]]
```

... SetDelayed: Tag Norm in Norm[vector_] is Protected.

```
Out[ ]:= $Failed
```

Die jeweiligen Tangentenvektoren sind daher

```
In[ ]:= T1un = D[rvec, u1]
```

```
T2un = D[rvec, u2]
```

```
T3un = D[rvec, z]
```

```
Out[ ]:= {a Cos[u2] Sinh[u1], a Cosh[u1] Sin[u2], 0}
```

```
Out[ ]:= {-a Cosh[u1] Sin[u2], a Cos[u2] Sinh[u1], 0}
```

```
Out[ ]:= {0, 0, 1}
```

```
In[*]:= h1 = Norm[T1un]
          h2 = Norm[T2un]
          h3 = Norm[T3un]
```

$$\text{Out[*]} = \sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}$$

$$\text{Out[*]} = \sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}$$

$$\text{Out[*]} = 1$$

```
In[*]:= T1 = T1un / h1
```

$$\text{Out[*]} = \left\{ \frac{a \cos[u2] \sinh[u1]}{\sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}}, \frac{a \cosh[u1] \sin[u2]}{\sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}}, 0 \right\}$$

$$\left\{ \frac{a \cos[u2] \sinh[u1]}{\sqrt{-\frac{1}{2} a^2 \cos[2 u2] + \frac{1}{2} a^2 \cosh[2 u1]}}, \frac{a \cosh[u1] \sin[u2]}{\sqrt{-\frac{1}{2} a^2 \cos[2 u2] + \frac{1}{2} a^2 \cosh[2 u1]}}, 0 \right\}$$

```
In[*]:= T2 = T2un / h2
```

$$\text{Out[*]} = \left\{ -\frac{a \cosh[u1] \sin[u2]}{\sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}}, \frac{a \cos[u2] \sinh[u1]}{\sqrt{\text{Abs}[a \cosh[u1] \sin[u2]]^2 + \text{Abs}[a \cos[u2] \sinh[u1]]^2}}, 0 \right\}$$

$$\left\{ -\frac{a \cosh[u1] \sin[u2]}{\sqrt{-\frac{1}{2} a^2 \cos[2 u2] + \frac{1}{2} a^2 \cosh[2 u1]}}, \frac{a \cos[u2] \sinh[u1]}{\sqrt{-\frac{1}{2} a^2 \cos[2 u2] + \frac{1}{2} a^2 \cosh[2 u1]}}, 0 \right\}$$

```
In[*]:= T3 = T3un / h3
```

$$\text{Out[*]} = \{0, 0, 1\}$$

$$\{0, 0, 1\}$$

(a)

Wir überprüfen die Orthogonalität:

```
In[*]:= T1.T2
```

$$\text{Out[*]} = 0$$

$$0$$

```
In[*]:= T1.T3
```

$$\text{Out[*]} = 0$$

$$0$$

```
In[*]:= T2.T3
```

$$\text{Out[*]} = 0$$

0

(b)

Das Volumenelement ist das Spatprodukt

$$\begin{aligned} & [\mathbf{T1unn} \cdot (\mathbf{T2unn} \times \mathbf{T3unn})] \, du_1 \, du_2 \, du_3 \\ & = h_1 \, h_2 \, h_3 \, du_1 \, du_2 \, du_3 \end{aligned}$$

und damit

In[*]:= **dV = Simplify[h1 h2 h3] du1 du2 du3**

Out[*]:= $du_1 \, du_2 \, du_3 \left(\text{Abs}[a \cosh[u_1] \sin[u_2]]^2 + \text{Abs}[a \cos[u_2] \sinh[u_1]]^2 \right)$

Wir können das auch noch in anderer Form darstellen:

In[*]:= **Simplify[dV /. {Cos[2 u2] -> 1 - 2 Sin[u2]^2,**
Cosh[2 u1] -> 1 + 2 Sinh[u1]^2}]

Out[*]:= $du_1 \, du_2 \, du_3 \left(\text{Abs}[a \cosh[u_1] \sin[u_2]]^2 + \text{Abs}[a \cos[u_2] \sinh[u_1]]^2 \right)$

Diese Rechnung ist übrigens nahezu identisch mit Hilfe der Jacobi-Determinante durchführbar:

In[*]:= **Coord = {u1, u2, z};**

JacMat = Table[D[rvec[[i]], Coord[[j]]], {i, 1, 3}, {j, 1, 3}]

Out[*]:= $\left\{ \left\{ a \cos[u_2] \sinh[u_1], -a \cosh[u_1] \sin[u_2], 0 \right\}, \right.$
 $\left. \left\{ a \cosh[u_1] \sin[u_2], a \cos[u_2] \sinh[u_1], 0 \right\}, \left\{ 0, 0, 1 \right\} \right\}$

In[*]:= **JacDet = Det[JacMat]**

Out[*]:= $a^2 \cosh[u_1]^2 \sin[u_2]^2 + a^2 \cos[u_2]^2 \sinh[u_1]^2$

Da dieser Ausdruck positiv ist, brauchen wir nicht nochmals den Betrag davon zu nehmen. Das ist auch schon das Ergebnis (=h1 h2 h3). Wir formen noch um, um die Identität zu sehen:

In[*]:= **Simplify[JacDet]**

Out[*]:= $-\frac{1}{2} a^2 (\cos[2 u_2] - \cosh[2 u_1])$

In[*]:= **ClearAll["Global`*"];**

In[*]:= **ClearAll["Global`*"];**

8.7

Ermitteln Sie für $\mathbf{r}(t)$ die Komponenten der Geschwindigkeit $\mathbf{v}$ im Basissystem der Kugelkoordinaten.

Lösungsweg

Man muss beachten, dass sich auch die Basisvektoren des Kugelkoordinatensystems zeitlich ändern können.

Der Ortsvektor im Kugelkoordinatensystem ist

$$\mathbf{r}(t) = r(t) \mathbf{e}_r(t)$$

damit finden wir

$$\mathbf{v}(\mathbf{t}) = \dot{r}(\mathbf{t}) \mathbf{e}_r(\mathbf{t}) + r(\mathbf{t}) \dot{\theta}(\mathbf{t}) \mathbf{e}_\theta(\mathbf{t}) + r(\mathbf{t}) \sin \theta(\mathbf{t}) \dot{\varphi}(\mathbf{t}) \mathbf{e}_\varphi(\mathbf{t})$$

Die Ableitung des Einheitsvektors steht senkrecht auf denselben, ist daher eine Linearkombination der beiden anderen Basisvektoren. In Aufgabe (8.5) wurde dies schon berechnet:

$$\mathbf{e}_r = \dot{\theta} \mathbf{e}_\theta + \sin \theta \dot{\varphi} \mathbf{e}_\varphi$$

Die Komponenten der Geschwindigkeit in der Kugelkoordinatenbasis sind daher

$$\mathbf{v}(\mathbf{t}) = (\dot{r}(\mathbf{t}), r(\mathbf{t}) \dot{\theta}(\mathbf{t}), r(\mathbf{t}) \sin \theta(\mathbf{t}) \dot{\varphi}(\mathbf{t}))$$

In[]:= `ClearAll["Global`*"];`

8.8

Welchen Wert hat das Volumenintegral

$$\left(\iiint_V \right) dV \sqrt{x_1^2 + x_2^2}$$

über den von den beiden Flächen


$$x_3 = x_1^2 + x_2^2 \quad \text{und} \quad x_3 = 9 - (x_1^2 + x_2^2)$$

eingeschlossenen Bereich?


Lösungsweg

Wie sehen die beiden Grenzflächen eigentlich aus?

In[]:= `PUp = Plot3D[x1^2 + x2^2, {x1, -3, 3}, {x2, -3, 3}]`



```
In[*]:= PDown = Plot3D[9 - (x1^2 + x2^2), {x1, -3, 3}, {x2, -3, 3}]
```




Die beiden Flächen durchdringen sich bei

$$x_1^2 + x_2^2 = 9 - (x_1^2 + x_2^2) \quad ,$$

$$\text{also } x_1^2 + x_2^2 = 9/2, \quad x_3 = 9/2.$$

```
In[*]:= Show[PUp, PDown]
```



Für die Rechnung gehen wir zu Zylinderkoordinaten über. Die Grenzflächen haben die Form

$$z_a = r^2 < z < z_b = 9 - r^2$$

Die Schnittlinie ist bei $r^2 = 9/2$.

Damit sind die Integrationsgrenzen

$$0 < \varphi \leq 2\pi, \quad 0 \leq r \leq 3/\sqrt{2}, \quad r^2 < z < 9 - r^2$$

Das Integrationsvolumenelement hat die Form

$$dV = r \, dr \, d\varphi \, dz$$

Die Integration lautet damit

$$\iiint d\varphi \, dr \, dz \, r^2$$

Über den Winkel kann man sofort integrieren. Weiter ergibt sich

```
In[*]:= Integrand = Integrate[2 π r^2, {z, r^2, 9 - r^2}]
```

```
Out[*]:= 2 π r^2 (9 - 2 r^2)
```

```
In[*]:= Ergebnis = Integrate[Integrand, {r, 0, 3/√2}]
```

```
Out[*]:= 81 √2 π / 5
```

```
In[*]:= ClearAll["Global`*"];
```

```
In[*]:= ClearAll["Global`*"];
```

8.9

Berechnen Sie

$$\left(\iint_B \right) dA \, \mathbf{n} \cdot \mathbf{F}$$

für das Vektorfeld $\mathbf{F} = (x_3, 2x_1, -4x_1^2x_3)$. B ist die Oberfläche eines Zylindermantels:


$$x_1^2 + x_2^2 = 25, \quad 0 \leq x_3 \leq 3$$

im ersten Oktanten. Verwenden Sie sowohl kartesische als auch Zylinderkoordinaten.

Lösungsweg

Die Oberfläche im ersten Oktanten hat das Aussehen

```
In[*]:= ParametricPlot3D[{5 Cos[phi], 5 Sin[phi], z}, {phi, 0, Pi/2}, {z, 0, 3}]
```



(a) Kartesische Koordinaten

In[*]:= **VecFeld** = {z, 2 x, -4 x^2 z}

Out[*]= {z, 2 x, -4 x^2 z}

Gekrümmter Teil BG:

Die Oberfläche wird durch x und z parametrisiert:

In[*]:= **rvec** = {x, Sqrt[25 - x^2], z};

Tx = D[rvec, x]

Tz = D[rvec, z]

Out[*]= {1, - $\frac{x}{\sqrt{25 - x^2}}$, 0}

Out[*]= {0, 0, 1}

Das Flächenelement **n** dA ergibt sich aus den Tangentenvektoren (im Parameterraum (x,z)) durch das äußere Produkt Tx×Tz zu

In[*]:= **Avec** = Cross[Tz, Tx]

Out[*]= { $\frac{x}{\sqrt{25 - x^2}}$, 1, 0}

Wir haben dabei dx dz nicht angeschrieben. Die Reihenfolge Tz, Tx wurde so gewählt, dass der Normalenvektor bei unserer Fläche nach außen zeigt. Der Integrand ist

In[*]:= **IntegrandBG** = Avec.VecFeld

Out[*]= $2x + \frac{xz}{\sqrt{25 - x^2}}$

Die Parameter haben die Grenzen $0 \leq x \leq 5$, $0 \leq z \leq 3$. Die Integration ergibt daher

In[*]:= **FBG** = Integrate[IntegrandBG, {z, 0, 3}, {x, 0, 5}]

Out[*]= $\frac{195}{2}$

(Substitutionsvorschlag: $u=25-x^2$.)

(b) Zylinderkoordinaten

Wir parametrisieren die Fläche mit Hilfe der Zylinderkoordinaten und haben daher

In[*]:= **rvec** = {5 Cos[phi], 5 Sin[phi], z};

Tphi = D[rvec, phi]

Tz = D[rvec, z]

Out[*]= {-5 Sin[phi], 5 Cos[phi], 0}

Out[*]= {0, 0, 1}

Das Flächenelement **n** dA ergibt sich aus den Tangentenvektoren (im Parameterraum (phi,z)) durch das äußere Produkt Tphi×Tz zu

In[*]:= **Avec** = Cross[Tphi, Tz]

Out[*]= {5 Cos[phi], 5 Sin[phi], 0}

Das Vektorfeld in Zylinderkoordinaten (aber nicht in der Basis der Zylinderkoordinaten!) ist

```
In[*]:= VecFeldZ = VecFeld /. {x -> 5 Cos[phi], y -> 5 Sin[phi]}
```

```
Out[*]:= {z, 10 Cos[phi], -100 z Cos[phi]^2}
```

Der Integrand ist dann

```
In[*]:= IntegrandBG = VecFeldZ.Avec
```

```
Out[*]:= 5 z Cos[phi] + 50 Cos[phi] Sin[phi]
```

und die Integration läuft über $0 \leq \phi \leq \pi/2$, $0 \leq z \leq 3$:

```
In[*]:= FBG = Integrate[IntegrandBG, {phi, 0, Pi/2}, {z, 0, 3}]
```

```
Out[*]:= 195/2
```

Das Ergebnis ist natürlich das gleiche.

(c) Zusatzaufgabe

Als Fleißaufgabe berechnen wir noch die anderen Oberflächenstücke des im ersten Oktanten eingeschlossenen Volumen:

die beiden Viertelkreise oben und unten (BU, BO)

und die beiden Rechtecke vorne und links (BV, BL).

Flächenstück BU:

...durch x, y parametrisiert (z=0). Der Normalvektor zeigt nach außen, daher hier nach unten (0,0,-1). Damit ist

```
In[*]:= IntegrandBU = {0, 0, -1}.VecFeld /. z -> 0
```

```
Out[*]:= 0
```

```
In[*]:= FBU = 0
```

```
Out[*]:= 0
```

Flächenstück BO:

..durch x, y parametrisiert (z=3). Der Normalvektor zeigt nach außen, daher hier nach oben (0,0,1).

Damit ist

```
In[*]:= IntegrandBO = {0, 0, 1}.VecFeld /. z -> 3
```

```
Out[*]:= -12 x^2
```

Die Integrationsgrenzen sind $0 \leq x \leq 5$, $0 \leq y \leq \sqrt{25 - x^2}$, und daher

```
In[*]:= FBO = Integrate[Integrate[IntegrandBO, {y, 0, Sqrt[25 - x^2]}], {x, 0, 5}]
```

```
Out[*]:= -1875 \pi/4
```

Flächenstück BV:

durch x,z parametrisiert. Der Normalvektor ist (0,-1,0).

```
In[*]:= IntegrandBV = {0, -1, 0}.VecFeld /. y -> 0
```

```
Out[*]:= -2 x
```

Die Integrationsgrenzen sind $0 \leq x \leq 5$, $0 \leq z \leq 3$, und daher

```
In[*]:= FBV = Integrate[IntegrandBV, {x, 0, 5}, {z, 0, 3}]
```

```
Out[*]:= -75
```

Flächenstück BL:

durch y,z parametrisiert. Der Normalvektor ist (-1,0,0).

```
In[*]:= IntegrandBL = {-1, 0, 0}.VecFeld /. x -> 0
```

```
Out[*]:= -z
```

Die Integrationsgrenzen sind $0 \leq y \leq 5$, $0 \leq z \leq 3$, und daher

```
In[*]:= FBL = Integrate[IntegrandBL, {y, 0, 5}, {z, 0, 3}]
```

```
Out[*]:= -45/2
```

Die Summe aller Beiträge ergibt sich zu

```
In[*]:= FBO + FBU + FBV + FBL + FBG
```

```
Out[*]:= -1875 π/4
```

Dieser Wert kann mit den Methoden des Kap. 9 auch durch ein Volumenintegral berechnet werden!

Uff. Das war nicht gerade wenig zu rechnen!

```
In[*]:= ClearAll["Global`*"];
```

```
In[*]:= ClearAll["Global`*"];
```

8.10

Bestimmen Sie (a) $\nabla \cdot \mathbf{F}$ und (b) $\nabla \times \mathbf{F}$ in kartesischen und in Zylinderkoordinaten und vergleichen Sie die Resultate. Für den Fall der Zylinderkoordinaten transformieren Sie zuerst $\mathbf{F}(x_1, x_2, x_3)$ in $\mathbf{F}(\rho, \varphi, x_3)$, und stellen Sie $\mathbf{F}$ im Basissystem $\mathbf{e}_\rho, \mathbf{e}_\varphi, \mathbf{e}_3$ dar.

$$\mathbf{F} = \left(x_1^2 x_2, 2 x_2 x_3, x_1 + x_3 \right).$$

Lösungsweg

(a) Kartesisches System

```
In[*]:= Fvec = {x1^2 x2, 2 x2 x3, x1 + x3}
```

```
Out[*]:= {x1^2 x2, 2 x2 x3, x1 + x3}
```

Im abgekürzten Verfahren:

```
In[*]:= Needs["VectorAnalysis`"];
```

```
In[*]:= SetCoordinates[Cartesian[x1, x2, x3]];
```

```
In[*]:= DivKart = Div[Fvec]
```

```
Out[*]:= 1 + 2 x1 x2 + 2 x3
```

Der Rotor ist daher:

In[*]:= RotKart = Curl[Fvec]

Out[*]:= $\{-2 x^2, -1, -x^2\}$

(b) Zylinderkoordinatensystem

Wir müssen zuerst **F** ins Basissystem der Zylinderkoordinaten transformieren. Die Einheitsvektoren dieses Systems (ausgedrückt in Komponenten im kartesischen) System sind

In[*]:= **e1** = {Cos[phi], Sin[phi], 0};

e2 = {-Sin[phi], Cos[phi], 0};

e3 = {0, 0, 1};

Die Komponenten von **F** im kartesischen System, aber ausgedrückt durch Zylinderkoordinaten, sind

In[*]:= **Fnew** = Fvec /. {x1 -> r Cos[phi], x2 -> r Sin[phi], x3 -> z}

Out[*]:= $\{r^3 \cos[\phi]^2 \sin[\phi], 2 r z \sin[\phi], z + r \cos[\phi]\}$

Die Komponenten im Zylinderkoordinatensystem ergeben sich durch Projektion des Vektors auf diese Basisvektoren (e1, e2, e3):

In[*]:= **Fzyl** = {Fnew.e1, Fnew.e2, Fnew.e3}

Out[*]:= $\{r^3 \cos[\phi]^3 \sin[\phi] + 2 r z \sin[\phi]^2, 2 r z \cos[\phi] \sin[\phi] - r^3 \cos[\phi]^2 \sin[\phi]^2, z + r \cos[\phi]\}$

Wir brauchen nun noch Divergenz und Rotor im Zylinderkoordinatensystem.

In[*]:= SetCoordinates[Cylindrical[r, phi, z]];

Zuerst zur Erinnerung die allgemeinen Ausdrücke:

In[*]:= Div[{F1[r, phi, z], F2[r, phi, z], F3[r, phi, z]}]

Out[*]:=
$$\frac{F1[r, \phi, z] + r F3^{(0,0,1)}[r, \phi, z] + F2^{(0,1,0)}[r, \phi, z] + r F1^{(1,0,0)}[r, \phi, z]}{r}$$

In[*]:= Curl[{F1[r, phi, z], F2[r, phi, z], F3[r, phi, z]}]

Out[*]:=
$$\left\{ \frac{-r F2^{(0,0,1)}[r, \phi, z] + F3^{(0,1,0)}[r, \phi, z]}{r}, F1^{(0,0,1)}[r, \phi, z] - F3^{(1,0,0)}[r, \phi, z], \frac{F2[r, \phi, z] - F1^{(0,1,0)}[r, \phi, z] + r F2^{(1,0,0)}[r, \phi, z]}{r} \right\}$$

Nun unsere Rechnung:

In[*]:= Div[Fzyl]

Out[*]:=
$$\frac{1}{r} \left(r + 2 r z \cos[\phi]^2 - r^3 \cos[\phi]^3 \sin[\phi] + 2 r^3 \cos[\phi] \sin[\phi]^3 + r (3 r^2 \cos[\phi]^3 \sin[\phi] + 2 z \sin[\phi]^2) \right)$$

Eine Vereinfachung liefert

In[*]:= DivZyl = Simplify[Div[Fzyl]]

Out[*]:= $1 + 2 z + r^2 \sin[2 \phi]$

Zum Vergleich schreiben wir das Ergebnis (a) aus kartesischen in Zylinder-Koordinaten um:

```
In[ ]:= DivKart /. {x1 -> r Cos[phi], x2 -> r Sin[phi], x3 -> z}
```

```
Out[ ]:= 1 + 2 z + 2 r^2 Cos[phi] Sin[phi]
```

Das stimmt offenbar überein ($\sin[2 \text{ phi}] = 2 \cos[\text{phi}] \sin[\text{phi}]$) !

Nun zum Rotor:

```
In[ ]:= RotZyl = Simplify[Curl[Fzyl]]
```

```
Out[ ]:= {- (1 + 2 r Cos[phi]) Sin[phi], -Cos[phi] + 2 r Sin[phi]^2, -r^2 Cos[phi]^2}
```

Wie sieht der Rotor im kartesischen System ausgedrückt durch Zylinderkoordinaten aus?

```
In[ ]:= RotKart /. {x1 -> r Cos[phi], x2 -> r Sin[phi], x3 -> z}
```

```
Out[ ]:= {-2 r Sin[phi], -1, -r^2 Cos[phi]^2}
```

Dies stimmt offenbar nicht mit RotZyl überein. Was ist passiert?

Achtung: Diese Komponenten sind im Zylinderkoordinatensystem und können nicht direkt mit den Komponenten im kartesischen System verglichen werden. Zum Vergleich mit den kartesischen Ergebnissen muss der entsprechende Rotor RotKart erst in das Zylinderkoordinatensystem projiziert werden:

```
In[ ]:= RotKartZyl = Simplify[{RotKart.e1, RotKart.e2, RotKart.e3} /.  
{x1 -> r Cos[phi], x2 -> r Sin[phi], x3 -> z}]
```

```
Out[ ]:= {- (1 + 2 r Cos[phi]) Sin[phi], -Cos[phi] + 2 r Sin[phi]^2, -r^2 Cos[phi]^2}
```

```
In[ ]:= ClearAll["Global`*"];
```

8.11

Die Fläche

$$3 x_1^2 + 3 x_2^2 + 5 x_3^2 = 15$$

begrenzt ein Rotationsellipsoid mit der Massendichte $\rho=1$. Wie groß ist das Trägheitsmoment des Körpers bei Rotation um die x_1 -Achse?

Lösungsweg

Zuerst verschaffen wir uns eine Vorstellung dieser Form. Das geht am besten in Kugelkoordinaten, also

```
In[ ]:= Simplify[3 x1^2 + 3 x2^2 + 5 x3^2 /.  
{x1 -> r Cos[phi] Sin[theta],  
x2 -> r Sin[phi] Sin[theta],  
x3 -> r Cos[theta]}]
```

```
Out[ ]:= r^2 (4 + Cos[2 theta])
```

Die Gleichung der Oberfläche ist also


$$r^2 (4 + \cos[2 \theta]) = 15$$

und mit $\cos[2 \theta] = 2 \cos[\theta]^2 - 1$ daher

$$r(\theta) = \sqrt{15 / (3 + 2 \cos[\theta]^2)}$$

und die graphische Darstellung ergibt

```
In[ ]:= SphericalPlot3D[Sqrt[15 / (3 +
                        2 Cos[theta]^2)],
                        {theta, 0, Pi},
                        {phi, 0, 2 Pi}]
```



Achtung: Dieser Ellipsoid dreht sich in diesem Beispiel nicht um die Symmetrieachse!

Version 1:

Wir verwenden Zylinderkoordinaten, deren Symmetrieachse die Drehachse (die x_1 -Achse) sein soll:

```
In[ ]:= SubstitutionList = {x1 -> z, x2 -> r Cos[phi],
                           x3 -> r Sin[phi]};
```

In diesem System hat die Begrenzungsfläche die Form


```
In[ ]:= Simplify[3 x1^2 + 3 x2^2 + 5 x3^2 /.
                SubstitutionList]
```

Out[]:= $4 r^2 + 3 z^2 - r^2 \cos[2 \phi]$

Für feste $z (=x_1)$ handelt es sich um Ellipsen:

$$15 - 3 z^2 = r^2 (4 - \cos[2 \phi])$$

```
In[ ]:= ParametricPlot[
  {Sqrt[15 / (4 - Cos[2 phi])] Cos[phi],
   Sqrt[15 / (4 - Cos[2 phi])] Sin[phi]},
  {phi, 0, 2 Pi}]
```



Das Differenzial des Trägheitsmoments ist $dI =$

$\rho r^2 dV$ (r = Abstand von der Drehachse) in unserem Beispiel also

$$dI = r^3 dr d\varphi dz$$

mit den Integrationsgrenzen

$$\sqrt{5} \leq z \leq \sqrt{5},$$

$$0 \leq \varphi \leq 2\pi,$$

$$0 \leq r \leq \sqrt{(15 - 3z^2) / (4 - \cos[2\varphi])}$$

Wir integrieren:

```
In[ ]:= I1 = Simplify[Integrate[r^3,
  {r, 0, Sqrt[(15 - 3 z^2) / (4 - Cos[2 phi])]}]]
```

$$\text{Out[]} = \frac{9(-5 + z^2)^2}{4(-4 + \cos[2\varphi])^2}$$

Wegen der Unabhängigkeit der Integrationsgrenzen könnten wir das Integral in zwei Faktoren (nur z -abhängig, nur φ -abhängig) zerlegen. Mit Mathematica ersparen wir uns das und integrieren einfach weiter:

```
In[ ]:= I2 = Simplify[Integrate[I1, {phi, 0, 2 Pi}]]
```

$$\text{Out[]} = \frac{2}{5} \sqrt{\frac{3}{5}} \pi (-5 + z^2)^2$$

Schließlich als Endergebnis:

```
In[ ]:= I3 = Simplify[Integrate[I2, {z, -√5, √5}]]
```

```
Out[ ]:= 
$$\frac{32 \pi}{\sqrt{3}}$$

```

Version 2:

Die Rechnung in den "normalen Zylinderkoordinaten" ist ebenfalls sehr einfach. Der Abstand von der Drehachse ist:

$$(\rho \sin[\varphi])^2 + z^2.$$

Die Integration ist:

```
In[ ]:= Integrate[( (ρ Sin[φ]) ^ 2 + z ^ 2) ρ, {φ, 0, 2 Pi}]
```

```
Out[ ]:= 
$$\pi \rho (2 z^2 + \rho^2)$$

```

```
In[ ]:= Integrate[π ρ (2 z^2 + ρ^2), {ρ, 0, √(15 - 5 z^2) / 3}]
```

```
Out[ ]:= 
$$\frac{25 \pi}{4} + \frac{5 \pi z^2}{6} - \frac{35 \pi z^4}{36}$$

```

```
In[ ]:= Integrate[ $\frac{25 \pi}{4} + \frac{5 \pi z^2}{6} - \frac{35 \pi z^4}{36}$ , {z, -√3, √3}]
```

```
Out[ ]:= 
$$\frac{5 \pi}{\sqrt{3}} + 9 \sqrt{3} \pi$$

```

```
In[ ]:= ClearAll["Global`*"];
```

Information zu Koordinatensystemen

In Mathematica kann das Koordinatensystem festgelegt werden, damit man dann den Gradienten (Grad), die Divergenz (Div), den Rotor (Curl) und den Laplace-Operator (Laplacian) berechnen kann.

```
In[ ]:= Needs["VectorAnalysis`"];
```

Man kann dann zum Beispiel folgende Befehle ausführen, um den Rotor eines Vektors zu berechnen:

```
In[ ]:= SetCoordinates[Cartesian[x1, x2, x3]];
```

```
In[ ]:= Curl[{x1 x2, x2 x3, x1 x3}]
```

```
Out[ ]:= {-x2, -x3, -x1}
```

Da dabei der Rechengang natürlicherweise nicht angegeben wird, werden in unseren Lösungsbeschreibungen Rechenschritte und Beispiele, die einfach nur die Berechnung dieser Differenzialoperationen sind, ebenfalls nicht angegeben.

```
In[ ]:= ClearAll["Global`*"];
```